EJEMPLO No 2 == PLACA RECTANGULAR TP = UN ELEMENTO 9 NODOS - v2018

Cell 12: Main program for verifying 9-node quad implementation with one element

1.- DATOS MALLA DE ELEMENTOS FINITOS

- NODOS

```
NodeCoordinates = N[{{0, 6}, {0, 3}, {0, 0}, {5 / 2, 6}, {5 / 2, 3},
   {5 / 2, 0}, {5, 6}, {5, 3}, {5, 0}}];
PrintPlaneStressNodeCoordinates[NodeCoordinates, "", {6, 4}];
```

node	x-coor	y-coor
1	0.0000	6.0000
2	0.0000	3.0000
3	0.0000	0.0000
4	2.5000	6.0000
5	2.5000	3.0000
6	2.5000	0.0000
7	5.0000	6.0000
8	5.0000	3.0000
9	5.0000	0.0000

```
numnod = Length[NodeCoordinates];
```

- ELEMENTOS

```
ElemNodes = {{1, 3, 9, 7, 2, 6, 8, 4, 5}};
```

```
numele = Length[ElemNodes];
```

```
ElemTypes = Table["Quad9", {numele}];
PrintPlaneStressElementTypeNodes[ElemTypes, ElemNodes, "", {}];
```

elem	type	node-list
1	Quad9	{1, 3, 9, 7, 2, 6, 8, 4, 5}

■ VISUALIZACION DE LA MALLA DE ELEMENTOS FINITOS

```
aspect = 6 / 5;
ProcessOptions = {True};
Plot2DElementsAndNodes[NodeCoordinates, ElemNodes, aspect,
  "One element mesh - 9-node quad", True, True];
```



2.- DATOS DEL PROBLEMA TENSION PLANA

■ MATERIAL

```
ClearAll[Em, ν, th];
Em = 10 000; ν = .25; Nsub = 4;
Emat = Em / (1 - ν^2) * {{1, ν, 0}, {ν, 1, 0}, {0, 0, (1 - ν) / 2}};
```

■ ASIGNACION DE MATERIAL Y ESPESOR A ELEMENTOS

```
th = 3;
```

```
ElemMaterials =  Table[Emat,  {numele}];
ElemFabrications = Table[th,   {numele}];
PrintPlaneStressElementMatFab[ElemMaterials, ElemFabrications, "", {}];
```

elem	material	fabrication
1	{{10666.7, 2666.67, 0.}, {2666.67, 10666.7, 0.}, {0., 0., 4000.}}	3

■ ASIGNACION DE CONDICIONES DE CONTORNO EN DESPLAZAMIENTOS

□ INICIALIZACION

```
NodeDOFValues = NodeDOFTags = Table[{0, 0}, {numnod}];
```

□ DEFINICION CONDICIONES DE CONTORNO EN DESPLAZAMIENTOS

```
NodeDOFValues[[1]] = NodeDOFValues[[7]] = {0, 25};
NodeDOFValues[[4]] = {0, 100};       (* nodal loads *)
NodeDOFTags[[1]] = NodeDOFTags[[2]] = {1, 0}; (* vroller @ nodes 1,2 *)
NodeDOFTags[[3]] = {1, 1};         (* fixed node 3 *)
NodeDOFTags[[6]] = NodeDOFTags[[9]] = {0, 1}; (* hroller @ nodes 6,9 *)
```

□ LISTADO DE CONDICIONES DE CONTORNO

```
PrintPlaneStressFreedomActivity[NodeDOFTags, NodeDOFValues, "", {}];
```

node	x-tag	y-tag	x-value	y-value
1	1	0	0.00	25.00
2	1	0	0.00	0.00
3	1	1	0.00	0.00
4	0	0	0.00	100.00
5	0	0	0.00	0.00
6	0	1	0.00	0.00
7	0	0	0.00	25.00
8	0	0	0.00	0.00
9	0	1	0.00	0.00

3.- SOLUCION DEL PROBLEMA Y VISUALIZACION DE RESULTADOS

■ SOLUCION DEL PROBLEMA

```
{NodeDisplacements, NodeForces, NodePlateCounts, NodePlateStresses,
  ElemBarNumbers, ElemBarForces} = PlaneStressSolution[
  NodeCoordinates, ElemTypes, ElemNodes,
  ElemMaterials, ElemFabrications,
  NodeDOFTags, NodeDOFValues, ProcessOptions];
```

■ IMPRESION DE RESULTADOS

```
PrintPlaneStressSolution[NodeDisplacements, NodeForces, NodePlateCounts,
   NodePlateStresses, "Computed Solution:", {}];
```

Computed Solution:

node	x-displ	y-displ	x-force	y-force	sigma-xx	sigma-yy	sigma-xy
1	0.0000	0.0060	0.0000	25.0000	0.0000	10.0000	0.0000
2	0.0000	0.0030	0.0000	0.0000	0.0000	10.0000	0.0000
3	0.0000	0.0000	0.0000	-25.0000	0.0000	10.0000	0.0000
4	-0.0006	0.0060	0.0000	100.0000	0.0000	10.0000	0.0000
5	-0.0006	0.0030	0.0000	0.0000	0.0000	10.0000	0.0000
6	-0.0006	0.0000	0.0000	-100.0000	0.0000	10.0000	0.0000
7	-0.0013	0.0060	0.0000	25.0000	0.0000	10.0000	0.0000
8	-0.0013	0.0030	0.0000	0.0000	0.0000	10.0000	0.0000
9	-0.0012	0.0000	0.0000	-25.0000	0.0000	10.0000	0.0000

4. - VISUALIZACION DE LOS DESPLAZAMIENTOS NODALES

CALCULO DE LOS VALORES MAXIMOS Y MINIMOS DE LOS DESPLAZAMIENTOS

```
ueps = 10.^(-3); nbands = 30;
ux = Table[NodeDisplacements[[n, 1]], {n, numnod}];
uy = Table[NodeDisplacements[[n, 2]], {n, numnod}];
{uxmax, uymax} = Abs[{Max[ux], Max[uy]}] + ueps;
{uxmin, uymin} = Abs[{Min[ux], Min[uy]}] + ueps;
uxmax = Max[uxmax, uxmin]; uxmin = -uxmax;
uymax = Max[uymax, uymin]; uymin = -uymax;
{uxinc, uyinc} = {uxmax - uxmin, uymax - uymin} / nbands;
```

VISUALIZACION DESPLAZAMIENTOS NODALES - X e Y

```
Print["uxmin,uxmax,uxinc=", {uxmin, uxmax, uxinc}];
ContourBandPlotNodeFuncOver2DMesh[NodeCoordinates, ElemNodes, ux, {uxmin, uxmax, uxinc},
 {True, True, True, False, True, True}, {2, 2}, aspect, "Displacement component ux"];
```

uxmin,uxmax,uxinc={-0.00225, 0.00225, 0.00015}




```
Print["uymin,uymax,uyinc=", {uymin, uymax, uyinc}];
ContourBandPlotNodeFuncOver2DMesh[NodeCoordinates, ElemNodes, uy, {uymin, uymax, uyinc},
 {True, True, True, False, True, True}, {2, 2}, aspect, "Displacement component uy"];
```

```
uymin,uymax,uyinc={-0.007, 0.007, 0.000466667}
```



5. - VISUALIZACION DE LAS TENSIONES - NODALES - NORMALES Y TANGENCIALES

CALCULO DE LOS VALORES MAXIMOS Y MINIMOS DE LAS TENSIONES NORMALES Y TANGENCIALES

```
sigeps = 10.^(-3); nbands = 30;
sxx = Table[NodePlateStresses[[n, 1]], {n, numnod}];
syy = Table[NodePlateStresses[[n, 2]], {n, numnod}];
sxy = Table[NodePlateStresses[[n, 3]], {n, numnod}];
{sxxmax, syymax, sxymax} = Abs[{Max[sxx], Max[syy], Max[sxy]}] + sigeps;
{sxxmin, syymin, sxymin} = Abs[{Min[sxx], Min[syy], Min[sxy]}] + sigeps;
sxxmax = Max[sxxmax, sxxmin]; sxxmin = -sxxmax;
syymax = Max[syymax, syymin]; syymin = -syymax;
sxymax = Max[sxymax, sxymin]; sxymin = -sxymax;
{sxxinc, syyinc, sxyinc} = {sxxmax - sxxmin, syymax - syymin, sxymax - sxymin} / nbands;
```

VISUALIZACION TENSIONES NODALES - NORMALES Y TANGENCIALES

```
Print["sxxmin,sxxmax,sxxinc=", {sxxmin, sxxmax, sxxinc}];
ContourBandPlotNodeFuncOver2DMesh[NodeCoordinates, ElemNodes, sxx, {sxxmin, sxxmax, sxxinc},
 {True, True, True, False, True, True}, {2, 2}, aspect, "Stress sigma-xx"];
```

```
sxxmin,sxxmax,sxxinc={-0.001, 0.001, 0.0000666667}
```




```
Print["syymin,syymax,syyinc=", {syymin, syymax, syyinc}];
ContourBandPlotNodeFuncOver2DMesh[NodeCoordinates, ElemNodes, syy, {syymin, syymax, syyinc},
 {True, True, True, False, True, True}, {2, 2}, aspect, "Stress sigma-yy"];
```

```
syymin,syymax,syyinc={-10.001, 10.001, 0.666733}
```




```
Print["sxymin,sxymax,sxyinc=", {sxymin, sxymax, sxyinc}];
ContourBandPlotNodeFuncOver2DMesh[NodeCoordinates, ElemNodes, sxy, {sxymin, sxymax, sxyinc},
 {True, True, True, False, True, True}, {2, 2}, aspect, "Stress sigma-xy"];
```

```
sxymin,sxymax,sxyinc={-0.001, 0.001, 0.0000666667}
```